

Functional Programming

Shin-Cheng Mu

FLOLAC 2026

0 To Begin With...

Prerequisites

If you have done the homework requested before this summer school, you should have familiarised yourself with

- values and types, and basic list processing,
- basics of type classes,
- defining functions by pattern matching,
- guards, case, local definitions by `where` and `let`,
- recursive definition of functions,
- and higher order functions.

Recommended Textbooks

- *Introduction to Functional Programming using Haskell* [Bir98]. My recommended book. Covers equational reasoning very well.
- *Programming in Haskell* [Hut16]. A thin but complete textbook.
- *Learn You a Haskell for Great Good!* [Lip11], a nice tutorial with cute drawings!
- *Real World Haskell* [OSG98].
- *Algorithm Design with Haskell* [BG20].

1 Definition and Proof by Induction

Total Functional Programming

- The next few lectures concerns inductive definitions and proofs of datatypes and programs.
- While Haskell provides allows one to define non-terminating functions, infinite data structures, for now we will only consider its total, finite fragment.
- That is, we temporarily

- consider only finite data structures,
- demand that functions terminate for all value in its input type, and
- provide guidelines to construct such functions.

- Infinite datatypes and non-termination can be modelled with more advanced theory, which we cannot cover in this course.

1.1 Induction on Natural Numbers

Recalling “Mathematical Induction”

- Let P be a predicate on natural numbers.
 - What is a predicate? Such a predicate can be seen as a function of type $\text{Nat} \rightarrow \text{Bool}$.
 - So far, we see Haskell functions as simple mathematical functions too.
 - However, Haskell functions will turn out to be more complex than mere mathematical functions later. To avoid confusion, we do not use the notation $\text{Nat} \rightarrow \text{Bool}$ for predicates.
- We’ve all learnt this principle of proof by induction: to prove that P holds for all natural numbers, it is sufficient to show that

0 $P\ 0$ holds;

1 $P\ (1 + n)$ holds provided that $P\ n$ does.

1.1.1 Proof by Induction

Proof by Induction on Natural Numbers

- We can see the above inductive principle as a result of seeing natural numbers as defined by the datatype ¹

`data Nat = 0 | 1+ Nat .`

- That is, any natural number is either 0, or $1+ n$ where n is a natural number.

¹Not a real Haskell definition.

- In this lecture, $\mathbf{1}_+$ is written in bold font to emphasise that it is a data constructor (as opposed to the function (+), to be defined later, applied to a number 1).

A Proof Generator

Given $P\ 0$ and $P\ n \Rightarrow P\ (\mathbf{1}_+\ n)$, how does one prove, for example, $P\ 3$?

$$\begin{aligned} & P\ (\mathbf{1}_+\ (\mathbf{1}_+\ (\mathbf{1}_+\ 0))) \\ \Leftarrow & \{ P\ (\mathbf{1}_+\ n) \Leftarrow P\ n \} \\ & P\ (\mathbf{1}_+\ (\mathbf{1}_+\ 0)) \\ \Leftarrow & \{ P\ (\mathbf{1}_+\ n) \Leftarrow P\ n \} \\ & P\ (\mathbf{1}_+\ 0) \\ \Leftarrow & \{ P\ (\mathbf{1}_+\ n) \Leftarrow P\ n \} \\ & P\ 0. \end{aligned}$$

Having done math. induction can be seen as having designed a program that generates a proof — given any $n :: Nat$ we can generate a proof of $P\ n$ in the manner above.

1.1.2 Inductively Definition of Functions

Inductively Defined Functions

- Since the type Nat is defined by two cases, it is natural to define functions on Nat following the structure:

$$\begin{aligned} exp & :: Nat \rightarrow Nat \rightarrow Nat \\ exp\ b\ 0 & = 1 \\ exp\ b\ (\mathbf{1}_+\ n) & = b \times exp\ b\ n. \end{aligned}$$

- Even addition can be defined inductively

$$\begin{aligned} (+) & :: Nat \rightarrow Nat \rightarrow Nat \\ 0 + n & = n \\ (\mathbf{1}_+\ m) + n & = \mathbf{1}_+\ (m + n). \end{aligned}$$

- Exercise: define (\times)?

A Value Generator

Given the definition of exp , how does one compute $exp\ b\ 3$?

$$\begin{aligned} & exp\ b\ (\mathbf{1}_+\ (\mathbf{1}_+\ (\mathbf{1}_+\ 0))) \\ = & \{ \text{definition of } exp \} \\ & b \times exp\ b\ (\mathbf{1}_+\ (\mathbf{1}_+\ 0)) \\ = & \{ \text{definition of } exp \} \\ & b \times b \times exp\ b\ (\mathbf{1}_+\ 0) \\ = & \{ \text{definition of } exp \} \\ & b \times b \times b \times exp\ b\ 0 \\ = & \{ \text{definition of } exp \} \\ & b \times b \times b \times 1. \end{aligned}$$

It is a program that generates a value, for any $n :: Nat$. Compare with the proof of P above.

Moral: Proving is Programming

An inductive proof is a program that generates a proof for any given natural number.

An inductive program follows the same structure of an inductive proof.

Proving and programming are very similar activities.

Without the $n + k$ Pattern

- Unfortunately, newer versions of Haskell abandoned the “ $n + k$ pattern” used in the previous slides:

$$\begin{aligned} exp & :: Int \rightarrow Int \rightarrow Int \\ exp\ b\ 0 & = 1 \\ exp\ b\ n & = b \times exp\ b\ (n - 1). \end{aligned}$$

- Nat is defined to be Int in `MiniPrelude.hs`. Without `MiniPrelude.hs` you should use Int .
- For the purpose of this course, the pattern $1+n$ reveals the correspondence between Nat and lists, and matches our proof style. Thus we will use it in the lecture.
- Remember to remove them in your code.

Proof by Induction

- To prove properties about Nat , we follow the structure as well.
- E.g. to prove that $exp\ b\ (m + n) = exp\ b\ m \times exp\ b\ n$.
- One possibility is to perform induction on m . That is, prove $P\ m$ for all $m :: Nat$, where $P\ m \equiv (\forall n :: exp\ b\ (m + n) = exp\ b\ m \times exp\ b\ n)$.

0 `goal` $P\ 0$.

`proof` For all n , we reason:

$$\begin{aligned} & exp\ b\ (0 + n) \\ = & \{ \text{defn. of } (+) \} \\ & exp\ b\ n \\ = & \{ \text{defn. of } (\times) \} \\ & 1 \times exp\ b\ n \\ = & \{ \text{defn. of } exp \} \\ & exp\ b\ 0 \times exp\ b\ n. \end{aligned}$$

1 `assume` $P\ m$.

`goal` $P\ (\mathbf{1}_+\ m)$.

proof For all n , we reason:

$$\begin{aligned}
 & \text{exp } b \ ((\mathbf{1}_+ m) + n) \\
 = & \quad \{ \text{defn. of } (+) \} \\
 & \text{exp } b \ (\mathbf{1}_+ (m + n)) \\
 = & \quad \{ \text{defn. of } \text{exp} \} \\
 & b \times \text{exp } b \ (m + n) \\
 = & \quad \{ \text{induction} \} \\
 & b \times (\text{exp } b \ m \times \text{exp } b \ n) \\
 = & \quad \{ (\times) \text{ associative} \} \\
 & (b \times \text{exp } b \ m) \times \text{exp } b \ n \\
 = & \quad \{ \text{defn. of } \text{exp} \} \\
 & \text{exp } b \ (\mathbf{1}_+ m) \times \text{exp } b \ n .
 \end{aligned}$$

Structure Proofs by Programs

- The inductive proof could be carried out smoothly, because both $(+)$ and exp are defined inductively on its lefthand argument (of type Nat).
- The structure of the proof follows the structure of the program, which in turns follows the structure of the datatype the program is defined on.

Lists and Natural Numbers

- We have yet to prove that (\times) is associative.
- The proof is quite similar to the proof for associativity of $(+)$, which we will talk about later.
- In fact, Nat and lists are closely related in structure.
- Most of us are used to think of numbers as atomic and lists as structured data. Neither is necessarily true.
- For the rest of the course we will demonstrate induction using lists, while taking the properties for Nat as given.

1.1.3 A Set-Theoretic Explanation of Induction

An Inductively Defined Set?

- For a set to be “inductively defined”, we usually mean that it is the *smallest* fixed-point of some function.
- What does that mean?

Fixed-Point and Prefixed-Point

- A *fixed-point* of a function f is a value x such that $f x = x$.
- **Theorem.** f has fixed-point(s) if f is a *monotonic function* defined on a complete lattice.
 - In general, given f there may be more than one fixed-point.
- A *prefixed-point* of f is a value x such that $f x \leq x$.
 - Apparently, all fixed-points are also prefixed-points.
- **Theorem.** the smallest prefixed-point is also the smallest fixed-point.

Example: Nat

- Recall the usual definition: Nat is defined by the following rules:
 0. 0 is in Nat ;
 1. if n is in Nat , so is $\mathbf{1}_+ n$;
 2. there is no other Nat .
- If we define a function F from sets to sets: $F X = \{0\} \cup \{\mathbf{1}_+ n \mid n \in X\}$, 0. and 1. above means that $F \text{Nat} \subseteq \text{Nat}$. That is, Nat is a prefixed-point of F .
- 2. means that we want the *smallest* such prefixed-point.
- Thus Nat is also the least (smallest) fixed-point of F .

Least Prefixed-Point

Formally, let $F X = \{0\} \cup \{\mathbf{1}_+ n \mid n \in X\}$, Nat is a set such that

$$F \text{Nat} \subseteq \text{Nat} , \tag{1}$$

$$(\forall X : F X \subseteq X \Rightarrow \text{Nat} \subseteq X) , \tag{2}$$

where (1) says that Nat is a prefixed-point of F , and (2) it is the least among all prefixed-points of F .

Mathematical Induction, Formally

- Given property P , we also denote by P the set of elements that satisfy P .
- That $P 0$ and $P n \Rightarrow P (\mathbf{1}_+ n)$ is equivalent to $\{0\} \subseteq P$ and $\{\mathbf{1}_+ n \mid n \in P\} \subseteq P$,
- which is equivalent to $F P \subseteq P$. That is, P is a prefixed-point!
- By (2) we have $\text{Nat} \subseteq P$. That is, all Nat satisfy P !
- This is “why mathematical induction is correct.”

Coinduction?

There is a dual technique called *coinduction* where, instead of least prefixed-points, we talk about *greatest postfixing points*. That is, largest x such that $x \leq f x$.

With such construction we can talk about infinite data structures.

1.2 Induction on Lists

Inductively Defined Lists

- Recall that a (finite) list can be seen as a datatype defined by:²

data $List\ a = [] \mid a : List\ a$.

- Every list is built from the base case $[]$, with elements added by $(:)$ one by one: $[1, 2, 3] = 1 : (2 : (3 : []))$.

All Lists Today are Finite

But what about infinite lists?

- For now let's consider finite lists only, as having infinite lists make the *semantics* much more complicated.³
- In fact, all functions we talk about today are total functions. No \perp involved.

Set-Theoretically Speaking...

The type $List\ a$ is the *smallest* set such that

- $[]$ is in $List\ a$;
- if xs is in $List\ a$ and x is in a , $x : xs$ is in $List\ a$ as well.

Inductively Defined Functions on Lists

- Many functions on lists can be defined according to how a list is defined:

$sum :: List\ Int \rightarrow Int$
 $sum\ [] = 0$
 $sum\ (x : xs) = x + sum\ xs$.

$map :: (a \rightarrow b) \rightarrow List\ a \rightarrow List\ b$
 $map\ f\ [] = []$
 $map\ f\ (x : xs) = f\ x : map\ f\ xs$.

- $sum\ [1..10] = 55$
- $map\ (1+) [1, 2, 3, 4] = [2, 3, 4, 5]$

²Not a real Haskell definition.

³What does that mean? Other courses in FLOLAC might cover semantics in more detail.

1.2.1 Append, and Some of Its Properties

List Append

- The function $(++)$ appends two lists into one

$(++) :: List\ a \rightarrow List\ a \rightarrow List\ a$
 $[] ++ ys = ys$
 $(x : xs) ++ ys = x : (xs ++ ys)$.

- Compare the definition with that of $(+)$!

Proof by Structural Induction on Lists

- Recall that every finite list is built from the base case $[]$, with elements added by $(:)$ one by one.
- To prove that some property P holds for all finite lists, we show that

0 $P\ []$ holds;

1 for all x and xs , $P\ (x : xs)$ holds provided that $P\ xs$ holds.

For a Particular List...

Given $P\ []$ and $P\ xs \Rightarrow P\ (x : xs)$, for all x and xs , how does one prove, for example, $P\ [1, 2, 3]$?

$P\ (1 : 2 : 3 : [])$
 $\Leftarrow \{ P\ (x : xs) \Leftarrow P\ xs \}$
 $P\ (2 : 3 : [])$
 $\Leftarrow \{ P\ (x : xs) \Leftarrow P\ xs \}$
 $P\ (3 : [])$
 $\Leftarrow \{ P\ (x : xs) \Leftarrow P\ xs \}$
 $P\ []$.

Appending is Associative

To prove that $xs ++ (ys ++ zs) = (xs ++ ys) ++ zs$.

Let $P\ xs = (\forall ys, zs :: xs ++ (ys ++ zs) = (xs ++ ys) ++ zs)$, we prove P by induction on xs .

0 **goal** $P\ []$.

proof For all ys and zs , we reason:

$[] ++ (ys ++ zs)$
 $= \{ \text{defn. of } (++) \}$
 $ys ++ zs$
 $= \{ \text{defn. of } (++) \}$
 $([] ++ ys) ++ zs$.

1 **assume** $P\ xs$

goal $P\ (x : xs)$

- *dropWhile* p xs drops the prefix from xs .

```
dropWhile      :: (a → Bool) → List a → List a
dropWhile p [] = []
dropWhile p (x : xs) | p x = dropWhile p xs
                    | otherwise = x : xs .
```

- Prove: *takeWhile* p xs ++ *dropWhile* p xs = xs .

- Some functions make more sense when it is defined only on non-empty lists:

```
f [x] = ...
f (x : xs) = ...
```

List Reversal

- *reverse* [1, 2, 3, 4] = [4, 3, 2, 1].

```
reverse      :: List a → List a
reverse []   = []
reverse (x : xs) = reverse xs ++ [x] .
```

- What about totality?

- They are in fact functions defined on a different datatype:

```
data List+ a = Singleton a | a : List+ a .
```

All Prefixes and Suffixes

- *inits* [1, 2, 3] = [[], [1], [1, 2], [1, 2, 3]]

```
inits      :: List a → List (List a)
inits []   = [[]]
inits (x : xs) = [] : map (x :) (inits xs) .
```

- We do not want to define *map*, *filter* again for *List⁺ a*. Thus we reuse *List a* and pretend that we were talking about *List⁺ a*.

- It's the same with *Nat*. We embedded *Nat* into *Int*.

- Ideally we'd like to have some form of *subtyping*. But that makes the type system more complex.

- *tails* [1, 2, 3] = [[1, 2, 3], [2, 3], [3], []]

```
tails      :: List a → List (List a)
tails []   = [[]]
tails (x : xs) = (x : xs) : tails xs .
```

Lexicographic Induction

Totality

- Structure of our definitions so far:

```
f [] = ...
f (x : xs) = ... f xs ...
```

- Both the empty and the non-empty cases are covered, guaranteeing there is a matching clause for all inputs.
- The recursive call is made on a “smaller” argument, guaranteeing termination.
- Together they guarantee that every input is mapped to some output. Thus they define *total* functions on lists.

- It also occurs often that we perform *lexicographic induction* on multiple arguments: some arguments decrease in size, while others stay the same.

- E.g. the function *merge* merges two sorted lists into one sorted list:

```
merge      :: List Int → List Int → List Int
merge [] [] = []
merge [] (y : ys) = y : ys
merge (x : xs) [] = x : xs
merge (x : xs) (y : ys) | x ≤ y = x : merge xs (y : ys)
                        | otherwise = y : merge (x : xs) ys .
```

1.2.3 Other Patterns of Induction

Variations with the Base Case

- Some functions discriminate between several base cases. E.g.

```
fib      :: Nat → Nat
fib 0    = 0
fib 1    = 1
fib (2 + n) = fib (1+n) + fib n .
```

Zip

Another example:

```
zip :: List a → List b → List (a, b)
zip [] [] = []
zip [] (y : ys) = []
zip (x : xs) [] = []
zip (x : xs) (y : ys) = (x, y) : zip xs ys .
```

Non-Structural Induction

- In most of the programs we've seen so far, the recursive call are made on direct sub-components of the input (e.g. $f(x : xs) = ..f xs..$). This is called *structural induction*.
 - It is relatively easy for compilers to recognise structural induction and determine that a program terminates.
- In fact, we can be sure that a program terminates if the arguments get “smaller” under some (well-founded) ordering.

Mergesort

- In the implementation of mergesort below, for example, the arguments always get smaller in size.

```
msort :: List Int → List Int
msort [] = []
msort [x] = [x]
msort xs = merge (msort ys) (msort zs) ,
  where n = length xs `div` 2
        ys = take n xs
        zs = drop n xs .
```

- What if we omit the case for $[x]$?
- If all cases are covered, and all recursive calls are applied to smaller arguments, the program defines a total function.

A Non-Terminating Definition

- Example of a function, where the argument to the recursive does not reduce in size:

```
f :: Int → Int
f 0 = 0
f n = f n .
```

- Certainly f is not a total function. Do such definitions “mean” something? We will talk about these later.

1.3 User Defined Inductive Datatypes

Internally Labelled Binary Trees

- This is a possible definition of internally labelled binary trees:

```
data ITree a = Null | Node a (ITree a) (ITree a) ,
```

- on which we may inductively define functions:

```
sumT :: ITree Nat → Nat
sumT Null = 0
sumT (Node x t u) = x + sumT t + sumT u .
```

Exercise: given $(\downarrow) :: Nat \rightarrow Nat \rightarrow Nat$, which yields the smaller one of its arguments, define the following functions

1. $minT :: Tree Nat \rightarrow Nat$, which computes the minimal element in a tree.
2. $mapT :: (a \rightarrow b) \rightarrow Tree a \rightarrow Tree b$, which applies the functional argument to each element in a tree.
3. Can you define (\downarrow) inductively on Nat ?⁴

Induction Principle for *Tree*

- What is the induction principle for *Tree*?
- To prove that a predicate P on *Tree* holds for every tree, it is sufficient to show that

0 P Null holds, and;

1 for every x, t , and u , if $P t$ and $P u$ holds, $P (\text{Node } x t u)$ holds.

- Exercise: prove that for all n and t , $minT (\text{mapT } (n+) t) = n + minT t$. That is, $minT \cdot \text{mapT } (n+) = (n+) \cdot minT$.

Induction Principle for Other Types

- Recall that $\text{data Bool} = \text{False} | \text{True}$. Do we have an induction principle for *Bool*?

- To prove a predicate P on *Bool* holds for all booleans, it is sufficient to show that

stmt0 $P \text{ False}$ holds, and

stmt1 $P \text{ True}$ holds.

- Well, of course.
- What about $(A \times B)$? How to prove that a predicate P on $(A \times B)$ is always true?

- One may prove some property P_1 on A and some property P_2 on B , which together imply P .

- That does not say much. But the “induction principle” for products allows us to extract, from a proof of P , the proofs P_1 and P_2 .

- Every inductively defined datatype comes with its induction principle.

- We will come back to this point later.

⁴In the standard Haskell library, (\downarrow) is called *min*.

2 Red-Black Tree

- A self-balancing binary search tree, often used to represent sets.
- Supports $O(\log n)$ -time searching, insertion, and deletion.
- One possible representation:

```
data RBTREE a = E |
    N Color (RBTREE a) a (RBTREE a) ,
data Color    = R | B .
```

Constraints

- It is a binary search tree.
 - In $N _ t \ x \ u$, every label in t is less than x , every label in u is more than x . The same holds for t and u .
- Each node is either colored red or black.
 - E is implicitly considered black.
- The root is black.
- Red nodes do not have red children.
- The number of black nodes from the root to each leaf is the same.

Searching

Searching in a red-black tree is just like that in a binary search tree:

```
search :: Int -> RBTREE Int -> Bool
search E           = False
search (N t x u) | k < x = ...
                  | k == x = ...
                  | k > x = ...
```

Exercise: what if we want to return the found element in a Maybe?

Insertion

- To insert a new element, perform a search to determine where to insert.
- The inserted node shall have color red.
- This would temporarily break the constraint that a red node shall not have a red children. We perform balancing upwards to restore the constraint. See the next slide.
- Finally we set the root to black.

Tree Balancing

- The re-balancing strategy is *not* unique.
- The strategy we will consider, shown in Figure 1, was presented by Okasaki [Oka99].
- Having only four rules, it is significantly simpler than those you'd find in most textbooks (which needs 8 rules or more)!
- Why?
- More will be discussed in the practicals.

3 Folds On Lists

A Common Pattern We've Seen Many Times...

```
sum [] = 0
sum (x : xs) = x + sum xs
```

```
length [] = 0
length (x : xs) = 1 + length xs
```

```
map f [] = []
map f (x : xs) = f x : map f xs
```

This pattern is extracted and called *foldr*:

```
foldr f e [] = e,
foldr f e (x : xs) = f x (foldr f e xs).
```

3.1 The Ubiquitous *foldr*

Replacing Constructors

```
foldr f e [] = e
foldr f e (x : xs) = f x (foldr f e xs)
```

- One way to look at *foldr* $(\oplus) e$ is that it replaces $[]$ with e and $(:)$ with (\oplus) :

```
foldr (\oplus) e [1, 2, 3, 4]
= foldr (\oplus) e (1 : (2 : (3 : (4 : []))))
= 1 \oplus (2 \oplus (3 \oplus (4 \oplus e)))
```

- $sum = foldr (+) 0$.
- $length = foldr (\lambda x n. 1 + n) 0$.
- $map f = foldr (\lambda x xs. f x : xs) []$.
- One can see that $id = foldr (:) []$.




Figure 1: Red-black tree balancing [Oka99].

Some Trivial Folds on Lists

- Function *max* returns the least upper bound of elements in a list:

$$\begin{aligned} \text{max } [] &= -\infty, \\ \text{max } (x : xs) &= x \uparrow \text{max } xs. \end{aligned}$$

$$\text{max} = \text{foldr } (\uparrow) -\infty.$$

- This function is actually called *maximum* in the standard Haskell Prelude, while *max* returns the maximum between its two arguments. For brevity, we denote the former by *max* and the latter by (\uparrow).

- Function *prod* returns the product of a list:

$$\begin{aligned} \text{prod } [] &= 1, \\ \text{prod } (x : xs) &= x \times \text{prod } xs. \end{aligned}$$

$$\text{prod} = \text{foldr } (\times) 1.$$

- Function *and* returns the conjunction of a list:

$$\begin{aligned} \text{and } [] &= \text{true}, \\ \text{and } (x : xs) &= x \wedge \text{and } xs. \end{aligned}$$

$$\text{and} = \text{foldr } (\wedge) \text{true}.$$

- Lets emphasise again that *id* on lists is a fold:

$$\begin{aligned} \text{id } [] &= [], \\ \text{id } (x : xs) &= x : \text{id } xs. \end{aligned}$$

$$\text{id} = \text{foldr } (:) [].$$

Some Functions We Have Seen...

- $(++)$ = *foldr* ($:$) *ys*.

$$\begin{aligned} (++) &:: [a] \rightarrow [a] \rightarrow [a] \\ [] ++ ys &= ys \\ (x : xs) ++ ys &= x : (xs ++ ys) . \end{aligned}$$

- concat* = *foldr* $(++)$ $[]$.

$$\begin{aligned} \text{concat} &:: [[a]] \rightarrow [a] \\ \text{concat } [] &= [] \\ \text{concat } (xs : xss) &= xs ++ \text{concat } xss . \end{aligned}$$

Replacing Constructors

- Understanding *foldr* from its type. Recall

$$\text{data } [a] = [] \mid a : [a] .$$

- Types of the two constructors: $[] :: [a]$, and $(:) :: a \rightarrow [a] \rightarrow [a]$.

More on Folds and Fold-fusion

- Compare the proof with the one yesterday. They are essentially the same proof.
- Fold-fusion theorem abstracts away the common parts in this kind of inductive proofs, so that we need to supply only the “important” parts.
- Tupling can be seen as a kind of fold-fusion. The derivation of *steepsum*, for example, can be seen as fusing:

$$\text{steepsum} \cdot \text{id} = \text{steepsum} \cdot \text{foldr } (:) [].$$

- Recall that $\text{steepsum } xs = (\text{steep } xs, \text{sum } xs)$. Reformulating *steepsum* into a fold allows us to compute it in one traversal.
- Not every function can be expressed as a fold. For example, $\text{tail} :: [a] \rightarrow [a]$ is not a fold!

3.3 More Useful Functions Defined as Folds

Longest Prefix

- The function call $\text{takeWhile } p \ xs$ returns the longest prefix of *xs* that satisfies *p*:

$$\begin{aligned} \text{takeWhile } p \ [] &= [] \\ \text{takeWhile } p \ (x : xs) &= \\ &\quad \text{if } p \ x \ \text{then } x : \text{takeWhile } p \ xs \\ &\quad \text{else } [] \end{aligned}$$

- E.g. $\text{takeWhile } (\leq 3) \ [1, 2, 3, 4, 5] = [1, 2, 3]$.
- It can be defined by a fold:

$$\begin{aligned} \text{takeWhile } p &= \text{foldr } (\text{tke } p) [], \\ \text{tke } p \ x \ xs &= \text{if } p \ x \ \text{then } x : xs \ \text{else } [] \end{aligned}$$

- Its dual, $\text{dropWhile } (\leq 3) \ [1, 2, 3, 4, 5] = [4, 5]$, is not a fold.

All Prefixes

- The function *inits* returns the list of all prefixes of the input list:

$$\begin{aligned} \text{inits } [] &= [[]], \\ \text{inits } (x : xs) &= [] : \text{map } (x :) (\text{inits } xs). \end{aligned}$$

- E.g. $\text{inits } [1, 2, 3] = [[]], [1], [1, 2], [1, 2, 3]$.
- It can be defined by a fold:

$$\begin{aligned} \text{inits} &= \text{foldr } \text{ini} \ [[]], \\ \text{ini } x \ xss &= [] : \text{map } (x :) \ xss. \end{aligned}$$

All Suffixes

- The function *tails* returns the list of all suffixes of the input list:

$$\begin{aligned} \text{tails } [] &= [[]], \\ \text{tails } (x : xs) &= \text{let } (ys : yss) = \text{tails } xs \\ &\quad \text{in } (x : ys) : yss. \end{aligned}$$

- E.g. $\text{tails } [1, 2, 3] = [[1, 2, 3], [2, 3], [3], []]$.
- It can be defined by a fold:

$$\begin{aligned} \text{tails} &= \text{foldr } \text{til} \ [[]], \\ \text{til } x \ (ys : yss) &= (x : ys) : yss. \end{aligned}$$

Scan

- $\text{scanr } f \ e = \text{map } (\text{foldr } f \ e) \cdot \text{tails}$.
- E.g.

$$\begin{aligned} \text{scanr } (+) \ 0 \ [1, 2, 3] &= \\ &= \text{map } \text{sum } (\text{tails } [1, 2, 3]) \\ &= \text{map } \text{sum } [[1, 2, 3], [2, 3], [3], []] \\ &= [6, 5, 3, 0]. \end{aligned}$$

- Of course, it is slow to actually perform $\text{map } (\text{foldr } f \ e)$ separately. By fold-fusion, we get a faster implementation:

$$\begin{aligned} \text{scanr } f \ e &= \text{foldr } (\text{sc } f) [e], \\ \text{sc } f \ x \ (y : ys) &= f \ x \ y : y : ys. \end{aligned}$$

4 Folds on Other Algebraic Datatypes

- Folds are a specialised form of induction.
- Inductive datatypes: types on which you can perform induction.
- Every inductive datatype give rise to its fold.
- In fact, an inductive type can be defined by its fold.

Fold on Natural Numbers

- Recall the definition:

$$\text{data } \text{Nat} = 0 \mid \mathbf{1}_+ \ \text{Nat} \ .$$

- Constructors: $0 :: \text{Nat}$, $(\mathbf{1}_+) :: \text{Nat} \rightarrow \text{Nat}$.
- What is the fold on *Nat*?

$$\begin{aligned} \text{foldN} &:: (a \rightarrow a) \rightarrow a \rightarrow \text{Nat} \rightarrow a \\ \text{foldN } f \ e \ 0 &= e \\ \text{foldN } f \ e \ (\mathbf{1}_+ \ n) &= f \ (\text{foldN } f \ e \ n) \ . \end{aligned}$$

Examples of $foldN$

- $(+n) = foldN (1+) n$.

$$\begin{aligned} 0 + n &= n \\ (1+ m) + n &= 1+ (m + n) . \end{aligned}$$

- $(\times n) = foldN (n+) 0$.

$$\begin{aligned} 0 \times n &= 0 \\ (1+ m) \times n &= n + (m \times n) . \end{aligned}$$

- $even = foldN not True$.

$$\begin{aligned} even 0 &= True \\ even (1+ n) &= not (even n) . \end{aligned}$$

Fold-Fusion for Natural Numbers

Theorem 2 ($foldN$ -Fusion). Given $f :: a \rightarrow a$, $e :: a$, $h :: a \rightarrow b$, and $g :: b \rightarrow b$, we have:

$$h \cdot foldN f e = foldN g (h e) ,$$

if $h (f x) = g (h x)$ for all x .

Exercise: fuse $even$ into $(+)$?

Folds on Trees

- Example: internally labelled binary tree:

```
data ITree a = Null
             | Node a (ITree a) (ITree a) .
```

- Fold for ITree:

$$\begin{aligned} foldIT :: (a \rightarrow b \rightarrow b \rightarrow b) \rightarrow b \rightarrow ITree a \rightarrow b \\ foldIT f e Null &= e \\ foldIT f e (Node a t u) &= \\ &f a (foldIT f e t) (foldIT f e u) . \end{aligned}$$

Folds on Trees

- Example: externally labelled binary tree:
- Some datatypes for trees:

```
data ETree a = Tip a
             | Bin (ETree a) (ETree a) .
```

- Fold for ETree:

$$\begin{aligned} foldET :: (b \rightarrow b \rightarrow b) \rightarrow (a \rightarrow b) \\ \rightarrow ETree a \rightarrow b \\ foldET f g (Tip x) &= g x \\ foldET f g (Bin t u) &= \\ &f (foldET f g t) (foldET f g u) . \end{aligned}$$

Some Simple Functions on Trees

- To compute the size of an ITree:

$$sizeIT = foldIT (\lambda x m n \rightarrow 1+ (m + n)) 0 .$$

- To sum up labels in an ETree:

$$sizeET = foldET (+) id .$$

- To compute a list of all labels in an ITree and an ETree:

$$\begin{aligned} flattenIT &= \\ &foldIT (\lambda x xs ys \rightarrow xs ++ [x] ++ ys) [] , \\ flattenET &= foldET (++) (\lambda x \rightarrow [x]) . \end{aligned}$$

- **Exercise:** what are the fusion theorems for $foldIT$ and $foldET$?

5 Program Derivation

5.1 Some Comments on Efficiency

Data Representation

- So far we have (surprisingly) been talking about mathematics without much concern regarding efficiency. Time for a change.
- Take lists for example. Recall the definition: **data** $List a = [] \mid a : List a$.
- Our representation of lists is biased. The left most element can be fetched immediately.
 - Thus. $(:)$, $head$, and $tail$ are constant-time operations, while $init$ and $last$ takes linear-time.
- In most implementations, the list is represented as a linked-list.

List Concatenation Takes Linear Time

- Recall $(++)$:

$$\begin{aligned} [] ++ ys &= ys \\ (x : xs) ++ ys &= x : (xs ++ ys) \end{aligned}$$

- Consider $[1, 2, 3] ++ [4, 5]$:

$$\begin{aligned} &(1 : 2 : 3 : []) ++ (4 : 5 : []) \\ &= 1 : ((2 : 3 : []) ++ (4 : 5 : [])) \\ &= 1 : 2 : ((3 : []) ++ (4 : 5 : [])) \\ &= 1 : 2 : 3 : ([] ++ (4 : 5 : [])) \\ &= 1 : 2 : 3 : 4 : 5 : [] \end{aligned}$$

- $(++)$ runs in time proportional to the length of its left argument.

Full Persistency

- Compound data structures, like simple values, are just values, and thus must be *fully persistent*.
- That is, in the following code:

```
let xs = [1, 2, 3]
    ys = [4, 5]
    zs = xs ++ ys
in ... body ...
```

- The *body* may have access to all three values. Thus ++ cannot perform a destructive update.

Linked v.s. Block Data Structures

- Trees are usually represented in a similar manner, through links.
- Fully persistency is easier to achieve for such linked data structures.
- Accessing arbitrary elements, however, usually takes linear time.
- In imperative languages, constant-time random access is usually achieved by allocating lists (usually called arrays in this case) in a consecutive block of memory.
- Consider the following code, where *xs* is an array (implemented as a block), and *ys* is like *xs*, apart from its 10th element:

```
let xs = [1..100]
    ys = update xs 10 20
in ... body ...
```

- To allow access to both *xs* and *ys* in *body*, the *update* operation has to duplicate the entire array.
- Thus people have invented some smart data structure to do so, in around $O(\log n)$ time.
- On the other hand, *update* may simply overwrite *xs* if we can somehow make sure that *nobody* other than *ys* uses *xs*.
- Both are advanced topics, however.

Another Linear-Time Operation

- Taking all but the last element of a list:

```
init [x] = []
init (x : xs) = x : init xs
```

- Consider *init* [1, 2, 3, 4]:

```
init (1 : 2 : 3 : 4 : [])
= 1 : init (2 : 3 : 4 : [])
= 1 : 2 : init (3 : 4 : [])
= 1 : 2 : 3 : init (4 : [])
= 1 : 2 : 3 : []
```

Sum, Map, etc

- Functions like *sum*, *maximum*, etc. needs to traverse through the list once to produce a result. So their running time is definitely $O(n)$, where n is the length of the list.
- If f takes time $O(t)$, *map f* takes time $O(n \times t)$ to complete. Similarly with *filter p*.
 - In a lazy setting, *map f* produces its first result in $O(t)$ time. We won't need lazy features for now, however.

5.2 Expand/Reduce Transformation

Sum of Squares

- Given a sequence a_1, a_2, \dots, a_n , compute $a_1^2 + a_2^2 + \dots + a_n^2$. Specification: *sumsq* = *sum* · *map square*.
- The spec. builds an intermediate list. Can we eliminate it?
- The input is either empty or not. When it is empty:

```
sumsq []
= { definition of sumsq }
  (sum · map square) []
= { function composition }
  sum (map square [])
= { definition of map }
  sum []
= { definition of sum }
  0
```

Sum of Squares, the Inductive Case

- Consider the case when the input is not empty:

```
sumsq (x : xs)
= { definition of sumsq }
  sum (map square (x : xs))
= { definition of map }
  sum (square x : map square xs)
= { definition of sum }
  square x + sum (map square xs)
= { definition of sumsq }
  square x + sumsq xs
```




Figure 2: How (+) allocates new (:) cells in the heap.

Alternative Definition for *sumsq*

- From $sumsq = sum \cdot map \ square$, we have proved that

$$\begin{aligned} sumsq [] &= 0 \\ sumsq (x : xs) &= square\ x + sumsq\ xs \end{aligned}$$

- Equivalently, we have shown that $sum \cdot map \ square$ is a solution of

$$\begin{aligned} f [] &= 0 \\ f (x : xs) &= square\ x + f\ xs \end{aligned}$$

- However, the solution of the equations above is unique.
- Thus we can take it as another definition of *sumsq*. Denotationally it is the same function; operationally, it is (slightly) quicker.
- Exercise: try calculating an inductive definition of *count*.

Remark: Why Functional Programming?

- Time to muse on the merits of functional programming. Why functional programming?
 - Algebraic datatype? List comprehension? Lazy evaluation? Garbage collection? These are just language features that can be migrated.
 - No side effects.⁵ But why taking away a language feature?
- By being pure, we have a simpler semantics in which we are allowed to construct and reason about programs.
 - In an imperative language we do not even have $f\ 4 + f\ 4 = 2 \times f\ 4$.
- Ease of reasoning. That's the main benefit we get.

⁵Unless introduced in disciplined ways. For example, through a monad.

Example: Computing Polynomial

Given a list $as = [a_0, a_1, a_2 \dots a_n]$ and $x :: Int$, the aim is to compute:

$$a_0 + a_1x + a_2x^2 + \dots + a_nx^n.$$

This can be specified by

$$poly\ x\ as = sum\ (zipWith\ (\times)\ as\ (iterate\ (\times x)\ 1)) ,$$

where *iterate* can be defined by

$$\begin{aligned} iterate &:: (a \rightarrow a) \rightarrow a \rightarrow List\ a \\ iterate\ f\ x &= x : map\ f\ (iterate\ f\ x) . \end{aligned}$$

Iterating a List

To get some intuition about *iterate* let us try expanding it:

$$\begin{aligned} iterate\ f\ x &= \{ \text{definition of } iterate \} \\ x : map\ f\ (iterate\ f\ x) &= \{ \text{definition of } map \} \\ x : map\ f\ (x : map\ f\ (iterate\ f\ x)) &= \{ map\ fusion \} \\ x : f\ x : map\ (f \cdot f)\ (iterate\ f\ x) &= \{ \text{definitions of } iterate\ \text{ and } map \} \\ x : f\ x : f\ (f\ x) : map\ (f \cdot f)\ (map\ f\ (iterate\ f\ x)) &= \{ map\ fusion \} \\ x : f\ x : f\ (f\ x) : map\ (f \cdot f \cdot f)\ (iterate\ f\ x) \dots \end{aligned}$$

Zippping with a Binary Operator

While *iterate* generate a list, it is immediately truncated by *zipWith*:

$$\begin{aligned} zipWith &:: (a \rightarrow b \rightarrow c) \rightarrow \\ &List\ a \rightarrow List\ b \rightarrow List\ c \\ zipWith\ (\oplus)\ [] &= [] \\ zipWith\ (\oplus)\ (x : xs)\ [] &= [] \\ zipWith\ (\oplus)\ (x : xs)\ (y : ys) &= \\ &x \oplus y : zipWith\ (\oplus)\ xs\ ys . \end{aligned}$$

Running the Specification

Try expanding $poly\ x\ [a, b, c, d]$, we get

$$\begin{aligned} & poly\ x\ [a, b, c, d] \\ = & sum\ (zipWith\ (\times)\ [a, b, c, d]\ (iterate\ (\times x)\ 1)) \\ = & \{ \text{expanding } iterate \} \\ & sum\ (zipWith\ (\times)\ [a, b, c, d] \\ & \quad (1 : (1 \times x) : (1 \times x \times x) : (1 \times x \times x \times x) : \\ & \quad \quad map\ (\times x)^4\ (iterate\ (\times x)\ 1))) \\ = & a + b \times x + c \times x \times x + d \times x \times x \times x . \end{aligned}$$

where f^4 denotes $f \cdot f \cdot f \cdot f$.

As the list gets longer, we get more $(\times x)$ accumulating. Can we do better?

The main calculation

$$\begin{aligned} & poly\ x\ (a : as) \\ = & \{ \text{definition of } poly \} \\ & sum\ (zipWith\ (\times)\ (a : as)\ (iterate\ (\times x)\ 1)) \\ = & \{ \text{definition of } iterate \} \\ & sum\ (zipWith\ (\times)\ (a : as) \\ & \quad (1 : map\ (\times x)\ (iterate\ (\times x)\ 1))) \\ = & \{ \text{definitions of } zipWith \text{ and } sum \} \\ & a + sum\ (zipWith\ (\times)\ as \\ & \quad (map\ (\times x)\ (iterate\ (\times x)\ 1))) \\ = & \{ \text{see below} \} \\ & a + sum\ (map\ (\times x)\ (zipWith\ (\times) \\ & \quad as\ (iterate\ (\times x)\ 1))) \\ = & \{ sum \cdot map\ (\times x) = (\times x) \cdot sum \} \\ & a + (sum\ (zipWith\ (\times)\ as\ (iterate\ (\times x)\ 1))) \times x \\ = & \{ \text{definition of } poly \} \\ & a + (poly\ x\ as) \times x . \end{aligned}$$

Zip-Map Exchange

In the 4th step we used the property $zipWith\ (\times)\ as \cdot map\ (\times x) = map\ (\times x) \cdot zipWith\ (\times)\ as$.

It applies to any operator (\otimes) that is associative. For an intuitive understanding:

$$\begin{aligned} & zipWith\ (\otimes)\ [a, b, c]\ (map\ (\otimes x)\ [d, e, f]) \\ = & [a \otimes (d \otimes x), b \otimes (e \otimes x), c \otimes (f \otimes x)] \\ = & \{ \text{associativity: } m \otimes (n \otimes k) = (m \otimes n) \otimes k \} \\ & [(a \otimes d) \otimes x, (b \otimes e) \otimes x, (c \otimes f) \otimes x] \\ = & map\ (\otimes x)\ (zipWith\ (\otimes)\ [a, b, c]\ [d, e, f]) . \end{aligned}$$

We can do a formal proof if we want.

Distributivity

In the 5th step we used the property $sum \cdot map\ (\times x) = (\times x) \cdot sum$. For that we need distributivity between addition and multiplication.

We used that law to push sum to the right.

This is the crucial property that allows us to speed up $poly$: we are allowed to factor out common $(\times x)$.

Computing Polynomial

To conclude, we get:

$$\begin{aligned} poly\ x\ [] &= 0 \\ poly\ x\ (a : as) &= a + (poly\ as) \times x , \end{aligned}$$

which uses a linear number of (\times) .

Let the Symbols Do the Work!

How do we know what laws to use or to assume?

By observing the form of the expressions. Let the symbols do the work.

5.3 Tupling

Steep Lists

- A *steep list* is a list in which every element is larger than the sum of those to its right:

$$\begin{aligned} steep &:: List\ Int \rightarrow Bool \\ steep\ [] &= True \\ steep\ (x : xs) &= steep\ xs \wedge x > sum\ xs . \end{aligned}$$

- The definition above, if executed directly, is an $O(n^2)$ program. Can we do better?
- Just now we learned to construct a generalised function which takes more input. This time, we try the dual technique: to construct a function returning more results.

Generalise by Returning More

- Recall that $fst\ (a, b) = a$ and $snd\ (a, b) = b$.
- It is hard to quickly compute $steep$ alone. But if we define

$$\begin{aligned} steepsum &:: List\ Int \rightarrow (Bool \times Int) \\ steepsum\ xs &= (steep\ xs, sum\ xs) , \end{aligned}$$

- and manage to synthesise a quick definition of $steepsum$, we can implement $steep$ by $steep = fst \cdot steepsum$.
- We again proceed by case analysis. Trivially,

$$steepsum\ [] = (True, 0) .$$

Deriving for the Non-Empty Case

For the case for non-empty inputs:

```

steepsum (x : xs)
= { definition of steepsum }
  (steep (x : xs), sum (x : xs))
= { definitions of steep and sum }
  (steep xs ∧ x > sum xs, x + sum xs)
= { extracting sub-expressions involving xs }
  let (b, y) = (steep xs, sum xs)
  in (b ∧ x > y, x + y)
= { definition of steepsum }
  let (b, y) = steepsum xs
  in (b ∧ x > y, x + y).

```

Synthesised Program

We have thus come up with a $O(n)$ time program:

```

steep          = fst · steepsum
steepsum []    = (True, 0)
steepsum (x : xs) = let (b, y) = steepsum xs
                    in (b ∧ x > y, x + y),

```

Being Quicker by Doing More?

- A more generalised program can be implemented more efficiently?
 - A common phenomena! Sometimes the less general function cannot be implemented inductively at all!
 - It also often happens that a theorem needs to be generalised to be proved. We will see that later.
- An obvious question: how do we know what generalisation to pick?
- There is no easy answer — finding the right generalisation one of the most difficulty act in programming!
- Sometimes we simply generalise by examining the form of the formula.

5.4 Accumulating Parameters

Reversing a List

- The function *reverse* is defined by:

```

reverse []      = [],
reverse (x : xs) = reverse xs ++ [x].

```

- E.g. $reverse [1, 2, 3, 4] = ((([] ++ [4]) ++ [3]) ++ [2]) ++ [1] = [4, 3, 2, 1]$.
- But how about its time complexity? Since $(++)$ is $O(n)$, it takes $O(n^2)$ time to revert a list this way.
- Can we make it faster?

5.4.1 Fast List Reversal

Introducing an Accumulating Parameter

- Let us consider a generalisation of *reverse*. Define:

```

revcat      :: List a → List a → List a
revcat xs ys = reverse xs ++ ys.

```

- If we can construct a fast implementation of *revcat*, we can implement *reverse* by:

```
reverse xs = revcat xs [].
```

Reversing a List, Base Case

Let us use our old trick. Consider the case when *xs* is $[]$:

```

revcat [] ys
= { definition of revcat }
  reverse [] ++ ys
= { definition of reverse }
  [] ++ ys
= { definition of (++) }
  ys.

```

Reversing a List, Inductive Case

Case $x : xs$:

```

revcat (x : xs) ys
= { definition of revcat }
  reverse (x : xs) ++ ys
= { definition of reverse }
  (reverse xs ++ [x]) ++ ys
= { since (xs ++ ys) ++ zs = xs ++ (ys ++ zs) }
  reverse xs ++ ([x] ++ ys)
= { definition of revcat }
  revcat xs (x : ys).

```

Linear-Time List Reversal

- We have therefore constructed an implementation of *revcat* which runs in linear time!

```

revcat [] ys      = ys
revcat (x : xs) ys = revcat xs (x : ys).

```

- A generalisation of *reverse* is easier to implement than *reverse* itself? How come?
- If you try to understand *revcat* operationally, it is not difficult to see how it works.
 - The partially reverted list is *accumulated* in *ys*.
 - The initial value of *ys* is set by $reverse xs = revcat xs []$.
 - Hmm... it is like a *loop*, isn't it?

5.4.2 Tail Recursion and Loops

Tracing Reverse

```
reverse [1, 2, 3, 4]
= revcat [1, 2, 3, 4] []
= revcat [2, 3, 4] [1]
= revcat [3, 4] [2, 1]
= revcat [4] [3, 2, 1]
= revcat [] [4, 3, 2, 1]
= [4, 3, 2, 1]

reverse xs      = revcat xs []
revcat [] ys    = ys
revcat (x : xs) ys = revcat xs (x : ys)

xs, ys ← XS, [];
while xs ≠ [] do
  xs, ys ← (tail xs), (head xs : ys);
return ys
```

Tail Recursion

- Tail recursion: a special case of recursion in which the last operation is the recursive call.

$$\begin{aligned} f x_1 \dots x_n &= \{\text{base case}\} \\ f x_1 \dots x_n &= f x'_1 \dots x'_n \end{aligned}$$

- To implement general recursion, we need to keep a stack of return addresses. For tail recursion, we do not need such a stack.
- Tail recursive definitions are like loops. Each x_i is updated to x'_i in the next iteration of the loop.
- The first call to f sets up the initial values of each x_i .

Accumulating Parameters

- To efficiently perform a computation (e.g. $\text{reverse } xs$), we introduce a generalisation with an extra parameter, e.g.:

$$\text{revcat } xs \ ys = \text{reverse } xs \ ++ \ ys.$$

- Try to derive an efficient implementation of the generalised function. The extra parameter is usually used to “accumulate” some results, hence the name.
 - To make the accumulation work, we usually need some kind of associativity.
- A technique useful for, but not limited to, constructing tail-recursive definition of functions.

Accumulating Parameter: Another Example

- Recall the “sum of squares” problem:

$$\begin{aligned} \text{sumsq } [] &= 0 \\ \text{sumsq } (x : xs) &= \text{square } x + \text{sumsq } xs. \end{aligned}$$

- The program still takes linear space (for the stack of return addresses). Let us construct a tail recursive auxiliary function.
- Introduce $\text{ssp } xs \ n = \text{sumsq } xs + n$.
- Initialisation: $\text{sumsq } xs = \text{ssp } xs \ 0$.
- Construct ssp :

$$\begin{aligned} \text{ssp } [] \ n &= 0 + n = n \\ \text{ssp } (x : xs) \ n &= (\text{square } x + \text{sumsq } xs) + n \\ &= \text{sumsq } xs + (\text{square } x + n) \\ &= \text{ssp } xs \ (\text{square } x + n). \end{aligned}$$

5.5 Conclusions

Conclusions

- Let the symbols do the work!
 - Algebraic manipulation helps us to separate the more mechanical parts of reasoning, from the parts that needs real innovation.
- For more examples of fun program calculation, see Bird [Bir10].
- For a more systematic study of algorithms using functional program reasoning, see Bird and Gibbons [BG20].

6 Monads and Effects

It is a misconception that functional languages do not allow side effects. In fact, many of them allow a variety of effects.

It is just that side effects must be introduced in a disciplined manner.

Disciplined? Such that we can use side effects, and still be able to reason about programs.

Side Effects

Anything a function does other than returning a value:

- reading/writing to a *mutable* variable,
- raising an exception,
- file/terminal I/O,
- asking for the current time,

- tossing a coin / generating a random number,
- partiality (possible failure),
- non-determinism,
- non-termination... and many more.

How to talk about all these effects?

Assume that we have a programming language supporting all these effects. Write an interpreter for this language, in a purely functional language.

Such an “interpreter” is actually a mathematical model of the language, offering a *semantics* for the language as well as the effects.

Example: A Pure Expression Evaluator

Our running example: consider the following type of expressions:

```
data Expr = Num Int | Neg Expr | Add Expr Expr .
```

How to evaluate an expression?

```
eval :: Expr -> Int
eval (Num n)    = n
eval (Neg e)    = -(eval e)
eval (Add e1 e2) = eval e1 + eval e2 .
```

6.1 A Language For Effects

- We are going to define a language that supports side effect. For now, imagine that is represented by some data structure.
- The type $m\ a$ denotes an effectful program which, when executed, may compute a pure value of type a after possibly performing some effects.
- A program having type $m\ a$ in Haskell can be executed by an interpreter. For now, imagine that it is a function $run :: m\ a \rightarrow a$ (we will extend this later).
- The type m is written in small letters, meaning that it is a type variable: we have not yet decided what it is.

Effects Are Marked by Types

Notice how we mark the existence of side effects by types.

- a denotes a pure value;
- $m\ a$ is an *effectful* computation that may return a value of type a .

Syntax for the Language

- Let us define a datatype for the syntax tree of the language. How should an effectful language look like?
- One might imagine a language having a syntax like this:

```
x ← comp1;
y ← comp2;
return (f x y) .
```

where $comp_1$ and $comp_2$ are computations that may be effectful, the value they compute are given to variables x and y , and the value of the entire computation is $f\ x\ y$.

- What about the types?
 - If $x :: a$ and $y :: b$, we must have $comp_1 :: m\ a$ and $comp_2 :: m\ b$.
 - If $f :: a \rightarrow b \rightarrow c$, the entire program has type $m\ c$.
 - The operator $return$ has type $c \rightarrow m\ c$.
- And what about the scope?
 - x can be used in line 2 and 3, but not in $comp_1$.
 - y can be used in line 3, but not in $comp_1$ and $comp_2$.

The Syntax in Haskell

- In Haskell, if we have the following “constructors” (the operator $(\ɾ)$ is pronounced “bind”):

```
return :: a -> m a
(&#x27E)   :: m a -> (a -> m b) -> m b ,
```

- we can build the program above by:

```
comp1 &#x27E (\x ->
comp2 &#x27E (\y ->
return (f x y))) .
```

- Haskell allows us to omit the parentheses:

```
comp1 &#x27E \x ->
comp2 &#x27E \y ->
return (f x y) .
```

- For now, we see the program above as data (a syntax tree), even though we write $return$ in small letters.

What Is This (\gg) All About?

Bind (\gg) is a bit like the “semicolon” in the syntax. It is also like an enhanced function application:

- $f \gg x$, where $f :: A \rightarrow B$ and $x :: A$, applies f to x .
- Recall that $m\ a$ denotes a *computation* that may yield a value of type a , while also incurs some side effects.
- $p \gg f$, where $f :: A \rightarrow m\ B$ and $p :: m\ A$, also “applies” f to p . However, evaluating p might incur some side effects. If the computation succeeds, we may extract some value of type A , which is passed to f , which in turn yields a computation $m\ B$.

Monad and Monad Laws

- If we intend to let *return* and (\gg) mean what we said, they should satisfy the following laws:

$$\begin{aligned} \text{left identity} & \quad \text{return } x \gg k = k\ x \text{ ,} \\ \text{right identity} & \quad mx \gg \text{return} = mx \text{ ,} \\ \text{associativity} & \quad (mx \gg k_1) \gg k_2 = \\ & \quad mx \gg (\lambda x \rightarrow k_1\ x \gg k_2) \text{ .} \end{aligned}$$

- For now, think of the equality above as “performing the same effects and return the same value(s)”.
- A type m and two operators *return* and (\gg) satisfying the laws above is called a *monad*.
 - There are other, equivalent ways to define monads.
 - Some people are scared off by this mathematical jargon. We wish to show in this lecture that monads need that be that hard.
- The laws are called *monad laws*.

The do Notation

To simplify and encourage the use of monads, Haskell provides a more concise notation, enclosed in the keyword **do**. We can actually write:

$$\begin{aligned} \text{do } x \leftarrow \text{comp}_1 \\ y \leftarrow \text{comp}_2 \\ \text{return } (f\ x\ y) \text{ .} \end{aligned}$$

Expression Evaluation, Monadically

Back to our *eval* example.

Instead of letting *eval* be a function $\text{Expr} \rightarrow \text{Int}$, let it *return a program that compute an Int*.

$$\begin{aligned} \text{eval} & :: \text{Expr} \rightarrow m\ \text{Int} \\ \text{eval } (\text{Num } n) & = \text{return } n \\ \text{eval } (\text{Neg } e) & = \text{eval } e \gg \lambda v \rightarrow \text{return } (-v) \\ \text{eval } (\text{Add } e_0\ e_1) & = \text{eval } e_0 \gg \lambda v_0 \rightarrow \\ & \quad \text{eval } e_1 \gg \lambda v_1 \rightarrow \\ & \quad \text{return } (v_0 + v_1) \text{ .} \end{aligned}$$

This is for possible extension later.

Note: this is not yet valid Haskell – some information is missing in the type.

Expression Evaluation In do Notation

Equivalently, in **do**-notation:

$$\begin{aligned} \text{eval} & :: \text{Expr} \rightarrow m\ \text{Int} \\ \text{eval } (\text{Num } n) & = \text{return } n \\ \text{eval } (\text{Neg } e) & = \text{do } v \leftarrow \text{eval } e \\ & \quad \text{return } (-v) \\ \text{eval } (\text{Add } e_0\ e_1) & = \text{do } v_0 \leftarrow \text{eval } e_0 \\ & \quad v_1 \leftarrow \text{eval } e_1 \\ & \quad \text{return } (v_0 + v_1) \text{ .} \end{aligned}$$

Which do you prefer?

Not Assignments!

It gives you an impression that you were writing an imperative program. Doesn't $v \leftarrow \text{eval } e$ look like “assign the value of *eval e* to variable *e*”?

In fact, $v \leftarrow \text{eval } e$ is closer to **let** in nature: it declares a new local variable v , whose scope extends to the end of the **do**-block. It can be shadowed by other bindings, like in **let**.

Translation

To be more precise, this is how a program using **do** is translated to monadic operators:

$$\begin{aligned} \text{do } \{ e \} & = e \\ \text{do } \{ e; es \} & = e \gg \lambda _ \rightarrow \text{do } \{ es \} \\ \text{do } \{ x \leftarrow e; es \} & = e \gg \lambda x \rightarrow \text{do } \{ es \} \\ \text{do } \{ \text{let } x = e; es \} & = \text{let } x = e \text{ in do } \{ es \} \text{ .} \end{aligned}$$

Monad Laws with do Notation

In this course, I prefer to see the (\gg) operator for reasoning. However, use of **do** notation is predominant in practical monadic programs.

The monad laws, for example, are written:

$$\begin{aligned} \text{do } \{ y \leftarrow \text{return } x; k\ y \} & = k\ x \text{ ,} \\ \text{do } \{ x \leftarrow mx; \text{return } x \} & = mx \text{ ,} \\ \text{do } \{ x \leftarrow mx; y \leftarrow k_1\ x; k_2\ y \} & = \\ \text{do } \{ y \leftarrow \text{do } \{ x \leftarrow mx; k_1\ x \}; k_2\ y \} & \text{ .} \end{aligned}$$

Which do you prefer?

6.2 Exceptions

What if we have division in Expr?

```
data Expr = Num Int | Neg Expr | Add Expr Expr
          | Div Expr Expr .
```

Division by zero should raise an exception.

Operators for Exception

We imagine having two more operators:

```
fail  :: m a ,
catch :: m a → m a → m a .
```

- *fail* raises an exception. In our current application it simply means division by zero.
- *catch m hdl* runs *m* and returns its value if it succeeds. If *m* fails, it runs *hdl*.

Note that they are still just the syntax of a program. The programs are not executed yet.

Evaluation with Failure

Now we define *eval* by:

```
eval (Num n)   = return n
eval (Neg e)   = eval e  » λv → return (-v)
eval (Add e0 e1) = eval e0 » λv0 →
                    eval e1 » λv1 →
                    return (v0 + v1)
eval (Div e0 e1) = eval e1 » λv1 →
                    if v1 == 0 then fail
                    else eval e0 » λv0 →
                    return (v0 `div` v1) .
```

Consecutive Product

How to multiply a sequence of numbers?

```
prod :: List Int → Int
prod []      = 1
prod (x : xs) = x * prod xs .
```

Hmm... can we stop early when there is a zero?

```
work :: List Int → m Int
work []      = return 1
work (0 : xs) = fail
work (x : xs) = work xs » λy →
                return (x * y) ,

fastprod :: List Int → m Int
fastprod xs = catch (work xs) (return 0) .
```

How do we show that *fastprod* is “correct”?

We want to prove that, for all *xs*,

```
fastprod xs = return (prod xs) .
```

Hmm... we need to know some more properties of *return*, (*»*), *fail* and *catch*.

Laws Regarding Exceptions

- For exceptions, we may want the following properties:

```
catch fail h = h ,
catch mx fail = m ,
catch mx (catch h h') = catch (catch m h) h' .
```

- Note that means *catch* and *fail* form a *monoid*.
- And this is how *catch* and *fail* interact with *return* and (*»*).

```
catch (return x) h = return x ,
fail » f = fail ,
```

- Looks reasonable... what about when *catch* meets (*»*)? Is it reasonable to demand that

```
catch mx h » f = catch (mx » f) (h » f) ?
```

Unfortunately no. See the practicals.

6.3 Reader

For the next example we want to allow expressions to have *let*-defined local variables, e.g.

```
let x = 3 in x + (let x = 4 in x) + (-x)
```

should evaluate to $3 + 4 + (-3) = 4$.

Extending Expr

To represent such expressions we extend the Expr type

```
data Expr = Num Int | Neg Expr | Add Expr Expr
          | Var Name | Let Name Expr Expr ,
```

where *type Name* = String.

The previous expression can be represented by:

```
Let "x" (Num 3)
  (Add (Add "x" (Let "x" (Num 4) (Var "x")))
        (Neg (Var "x")))
```

Environment

So, what is the value of $x + 2$?

We don't know, unless we know the value of *x*.

An *environment* is a mapping from variables to values. For now, we denote it by:

```
type Env = [(Name, Int)] .
```

We can also define a function *lookup* :: Env → Maybe Int.

“Having an Environment” as an Effect

- “Having an environment” can be seen as an effect.
- This effect is more commonly called the “Reader” effect, meaning that we can read from an environment.
- We extend our effect language with two operators:

$$\begin{aligned} ask &:: m \text{ Env} , \\ local &:: (\text{Env} \rightarrow \text{Env}) \rightarrow m \ a \rightarrow m \ a . \end{aligned}$$

- ask returns the environment.
- if the current environment is env , $local \ f \ m$ performs m in a local environment $f \ env$.

Evaluating an Expression Given an Environment

The usual cases stay the same.

$$\begin{aligned} eval &:: \text{Expr} \rightarrow m \ \text{Int} \\ eval \ (\text{Num } n) &= return \ n \\ eval \ (\text{Neg } e) &= eval \ e \ \gg \ \lambda v \rightarrow return \ (-v) \\ eval \ (\text{Add } e_0 \ e_1) &= eval \ e_0 \ \gg \ \lambda v_0 \rightarrow \\ &\quad eval \ e_1 \ \gg \ \lambda v_1 \rightarrow \\ &\quad return \ (v_0 + v_1) . \end{aligned}$$

Cases relevant to variables and `let`:

$$\begin{aligned} eval \ (\text{Var } x) &= \\ &ask \ \gg \ \lambda env \rightarrow \\ &\quad \text{case } lookup \ env \ x \ \text{of } \text{Just } v \rightarrow return \ v \\ eval \ (\text{Let } x \ e_0 \ e_1) &= \\ &eval \ e_0 \ \gg \ \lambda v \rightarrow \\ &\quad local \ ((x, v):) \ (eval \ e_1) . \end{aligned}$$

Laws Regarding Readers

For readers, we may want the following properties. Firstly, regarding ask ,

$$\begin{aligned} ask \ \gg \ \lambda v \rightarrow return \ e &= return \ e , \\ ask \ \gg \ \lambda v_0 \rightarrow ask \ \gg \ \lambda v_1 \rightarrow f \ v_0 \ v_1 &= \\ ask \ \gg \ \lambda v \rightarrow f \ v \ v . \end{aligned}$$

Secondly, regarding $local$:

$$\begin{aligned} local \ g \ (return \ e) &= return \ e , \\ local \ g \ (p \ \gg \ f) &= local \ g \ p \ \gg \ (local \ g \cdot f) . \end{aligned}$$

Finally, when $local$ meets ask :

$$local \ g \ ask = ask \ \gg \ (return \cdot g) .$$

What About Missing Variables?

- But wait... what if $lookup$ cannot find the variable?
- Our program has to return an exception.
- Therefore we actually need a monad that allows two effects: exception, and reader.

6.4 Implementing the Exception and Reader Effect

Separation of Concerns

The laws are used to reason about monadic programs — assuming that the monad exists and obey the laws.

Independently, we design interpreters for the effect and implement the methods, ensuring that they satisfy the laws.

Running a Program With Exceptions

- Assume that exception is the only effect. How do you run a program?
- Let us use the Maybe type! If a program $m \ a$ successfully computes some x , we return `Just x`. Otherwise we return `Nothing`.

$$\begin{aligned} run &:: m \ a \rightarrow \text{Maybe } a \\ run \ (return \ x) &= \text{Just } x \\ run \ (m \ \gg \ f) &= \text{case } run \ m \ \text{of} \\ &\quad \text{Just } x \rightarrow f \ x \\ &\quad \text{Nothing} \rightarrow \text{Nothing} \\ run \ fail &= \text{Nothing} \\ run \ (catch \ m \ hdl) &= \dots \end{aligned}$$

Skip the Intermediate Layer

- But we immediately see that the intermediate layer is redundant. What if we just let $m = \text{Maybe}$?
- The operators can be implemented by:

$$\begin{aligned} return \ x &= \text{Just } x , \\ \text{Just } x \ \gg \ f &= f \ x \\ \text{Nothing} \ \gg \ f &= \text{Nothing} , \\ fail &= \text{Nothing} , \\ catch \ (\text{Just } x) \ hdl &= \text{Just } x \\ catch \ \text{Nothing} \ hdl &= \text{Nothing} . \end{aligned}$$

- Then we simply have $run = id$.
- Of course, we are obliged to show that the implementation above satisfies all the laws we demanded.

Evaluating a Program Needing an Environment

- Now assume that reader is the only effect.
- To execute a program needing an environment, you need an environment. Therefore, *run* has type:

$$run :: m\ a \rightarrow Env \rightarrow a .$$

- That is, *run* takes a program, and an environment, and computes a result having type *a*.
- Or, the result of executing a program *m a* is a function $Env \rightarrow a$.
- Again, what if we let $m\ a = Env \rightarrow a$?
- The operators can be implemented by:

```
return :: a → m a
return x env = x ,
(⊗) :: m a → (a → m b) → m b
(mx ⊗ f) env = let y = mx env in f y env ,
ask :: m Env
ask env = env ,
local :: (Env → Env) → m a → m a
local f mx env = mx (f env) .
```

Evaluating a Program with Exception and Reader

- What about a program that uses both exception and reader?
- To run a program, we need an environment. And we return a *Maybe*.

$$run :: m\ a \rightarrow Env \rightarrow Maybe\ a .$$

- Again, we can simply let $m\ a = Env \rightarrow Maybe\ a$.
- Try implementing *return*, (\otimes), *ask*, *local*, *fail*, and *catch*?

Combining Effects

When a monad implements multiple effects, it is supposed to

- satisfy laws of all the effects, as well as
- “tensor” of these effects, meaning that operators from each effects commute with each other.

Having chosen some effects, how do we implement a monad that meets the above requirements?

Commutivity of Effects

For example, when operators of reader meet *fail*:

$$ask \otimes \lambda v \rightarrow fail = fail ,$$
$$local\ f\ fail = fail .$$

When operators of reader meet *catch*:

$$ask \otimes \lambda v \rightarrow catch\ mx\ my =$$
$$catch\ (ask \otimes \lambda v \rightarrow mx)$$
$$(ask \otimes \lambda v \rightarrow my) ,$$
$$local\ f\ (catch\ mx\ my)$$
$$catch\ (local\ f\ mx)\ (local\ f\ my) .$$

Composing Monads?

It is known that monads are difficult to compose, in the sense that once two monads are implemented, it is hard to combine them to form a monad having both their effects.

People have come up with various methods, e.g. *monad transformers*[LHJ95], and *effect handling*[K115], which we cannot cover in this course.

However, properties of effects are easy to compose: just take the union (and “tensor”) of all their properties.

6.5 The Haskell Reality...

- The code we presented so far are *not* actual, executable Haskell code.
- Haskell uses *type classes* allow several datatypes to share the same names of operators.
- You do not need to understand type classes to understand monads and effects. It is actually easier to do without them when we calculate on paper.
- In an actual program, however, it is convenient using the type class mechanism.

The Monad Typeclass

Generally speaking, a type constructor *m* and two operators *return* and (\otimes) constitute a *monad*:

```
class Monad m where
  return :: a → m a
  (⊗)    :: m a → (a → m b) → m b .
```

Note: this is still a simplification. In the standard Haskell, *Monad* is a subclass of *Applicative*... but we will avoid introducing the whole infrastructure here.

A Type Class For Failable Monads

Now assume that we want the monad to support *fail* and *catch*. To be abstract, we do not assume what datatype it is, but declare types that support these operators as classes:

```
class Monad m => MonadFail m where
  fail :: m a
class MonadFail m => MonadCatch m where
  catch :: m a -> m a -> m a
```

Instance Declaration for Maybe

Somewhere in the library, one may declare that *Maybe* is instances of the classes above:

```
instance Monad Maybe where
  return x = ...
  mx >> f = ...
instance MonadFail Maybe where
  fail = Nothing ,
instance MonadCatch Maybe where
  catch = ...
```

A Type Class For Readers

Any monad *m* that supports *ask* and *local* is in the class *MonadReader*:

```
class Monad m => MonadReader env m where
  ask :: m env
  local :: (env -> env) -> m a -> m a ,
```

where *env* is the type of environments — now a parameter, to be adapted to different use cases.

Instance Declaration

Similarly, we let

```
type Reader env a = env -> a .
```

Ideally, we wish to define

```
instance Monad (Reader env) where ...
instance MonadReader (Reader env) where
  ask env      = env
  local f mx env = mx (f env) .
```

Instance Declaration... In Reality

However, the type system of Haskell has a limitation that type synonyms cannot be instances of type classes.

Therefore we have to wrap $env \rightarrow a$ in a **data** definition:⁶

⁶In fact people generally use a **newtype** in this case, which is different from **data** in subtle ways. But we choose not to complicate the matter.

```
data Reader env a = Rdr (env -> a)
instance Monad (Reader e) where
  return x    = Rdr (\env -> x)
  Rdr mx >> f = Rdr (\env -> f (mx env) env) .
```

Alternative Implementations

Maybe is not the only implementation of *MonadFail* and *MonadCatch*, and *Reader* is not the only implementation of *MonadReader*.

Whatever the implementation is, it should satisfy the relevant laws.

Class Constraints in Types

The actual type of *eval* will be

```
eval :: (Monad m, MonadReader m, MonadFail m) =>
  Expr -> m Int ,
```

meaning that we can use any datatype that supports the operators of *Monad*, *MonadReader*, and *MonadFail*.

Note On The Use Of Type Classes

Two things to note:

- Type classes are *not* necessary to talk about monads! We use type class just to make it clear that our discussions are not tied to a particular implementation.
- Our type classes are different from those in standard Haskell Prelude, but closer to that of Gibbons and Hinze [GH11].

6.6 State

Recall that in $x \leftarrow m$ in the **do**-notation, *x* is *not* a mutable variable. That is covered by the *state* effect.

In the *state* effect, we have *one*, *anonymous* global mutable variable of type *s*, and two methods:

- *get* :: *m s* retrieves the value of the mutable variable;
- *put* :: *s -> m ()* assigns a value to the mutable variable.

The “Semicolon”

Note that *put* returns *()*, since *put* itself does not yield information. We use it merely for its side effect.

The value returned by *put* can be discarded.

Since it happens a lot we define a variation of (\gg):

```
(>) :: m a -> m b -> m b
mx > my = mx >> \_ -> my .
```

It is like a “semicolon” in imperative programs.

Laws For *get* and *put*

It should be reasonable to demand that they satisfy the following laws:

```

get-put      get ≧ put = return () ,
put-get      put e ≧ get = put e ≧ return e ,
put-put      put e0 ≧ put e1 = put e1 ,

```

and a law similar to that of *ask*:

```

get-get      get ≧ λv0 →
                get ≧ λv1 → f v0 v1 =
                get ≧ λv → f v v

```

Reasoning about Stateful Programs

With these laws we can already reason about programs that manipulate states.

See the practicals.

Implementation of State

And how do we implement the state effect?

“A program that returns a value of type *a* while being able to read from and write to a mutable variable of type *s*” can be represented by a function $s \rightarrow (a, s)$.

- Like reader, the function takes the initial value of the variable as its input;
- unlike reader, it returns not just an *a*, but also *the new value of the mutable variable*.

Implementing *return* and (\gg) for State

Let $m\ a = s \rightarrow (a, s)$ (for some fixed *s*). How do we implement the monad operators for state? Try it yourself before checking the answers below.!

```

return :: a → m a
return x s = (x, s) ,
(≫) :: m a → (a → m b) → m b
(mx ≧ k) s0 = let (y, s1) = mx s0
                in k y s1 .

```

Implementing *get* and *put*

And how do we implement *get* and *put*?

```

get :: m s
get s = (s, s) ,
put :: s → m ()
put s1 s0 = ((, s1) .

```

But of course, it is only one of the possible implementations.

In Reality...

Again, due to the limitation of Haskell’s type system, the real code is not that sleek...

```

data State s a = St (s → (a, s)) .
instance Monad (State s) where
    return x = St ...
    St mx ≧ k = St ...

```

Try finishing them yourself!

6.7 Nondeterminism

A pure function maps an input to exactly one output.

Not producing an output is an effect.

Possibly producing more than one output is also an effect.

A Type Class for Nondeterminism

Recall:

```

class Monad m ⇒ MonadFail m where
    fail :: m a

```

We define:

```

class Monad m ⇒ MonadAlt m where
    (⋄) :: m a → m a → m a
class (MonadFail m, MonadAlt m) ⇒
    MonadNondet m where

```

Let $mx \diamond my$ denote “either *mx* or *my*”.

Laws for Nondeterminism

What laws can we demand? We usually want *fail* and (\diamond) to form a monoid:

```

fail ⋄ mx = mx = mx ⋄ fail ,
(mx ⋄ my) ⋄ mz = mx ⋄ (my ⋄ mz) .

```

We probably wish that the order and numbers of choices we make do not matter:

```

idempotency: mx ⋄ mx = mx ,
commutivity: mx ⋄ my = my ⋄ mx .

```

Bind distributes into (\diamond) ... from both directions!

```

(mx ⋄ my) ≧ f =
    (mx ≧ f) ⋄ (my ≧ f) ,
mx ⋄ (λx → f x ⋄ g x) =
    (mx ≧ f) ⋄ (mx ≧ g) .

```

However, most implementations do not satisfy all the laws.

Lists as MonadNondet

If nondeterminism is the only effect we need, ideally, we can implement nondeterminism by sets.

Unfortunately we do not have real sets in Haskell. One can simulate nondeterminism using lists.

```
instance MonadFail List where
```

```
fail = []
```

```
instance MonadAlt List where
```

```
( $\phi$ ) = (++)
```

```
instance MonadNondet List where
```

Note: in fact, Haskell does not allow us to say List here. We have to use [].

However, the implementation above does not satisfy idempotency and commutativity.

An “honest” representation of sets requires Eq *a* and incurs certain overhead. As a trade off, we may keep using List as long as we remember that the results we get are only correct “modulo idempotency and commutivity”.

Nondeterminism and State

Can we have a monad that support both nondeterminism and state? It is supposed to satisfy laws of both effects.

There are at least two possible implementations:

```
type StNd s a = s  $\rightarrow$  [(a, s)] ;
```

```
type StNd s a = s  $\rightarrow$  ([a], s) .
```

What are their differences?

The latter does not satisfy the second distributivity law. However, there are cases when we find it useful.

6.8 Some Final Words

There are a lot about monads we cannot cover here.

- Other alternative, equivalent definition of monads (e.g. in terms of *fmap*, *return*, and *join*), and their roles in category theory.
- Relationship with other important constructs, e.g. applicative functors, monoids...

Our presentation of monads is a simplified one, different from that in Haskell Prelude, which is more tightly coupled with the “big picture.”

Currently, combining and managing effects is still an active research area.

Our focus is on proving properties of monadic programs. However, this is a relatively undeveloped field.

References

- [BG20] Richard S. Bird and Jeremy Gibbons. *Algorithm Design with Haskell*. Cambridge University Press, 2020.
- [Bir98] Richard S. Bird. *Introduction to Functional Programming using Haskell*. Prentice Hall, 1998.
- [Bir10] Richard S. Bird. *Pearls of Functional Algorithm Design*. Cambridge University Press, 2010.
- [GH11] Jeremy Gibbons and Ralf Hinze. Just do it: simple monadic equational reasoning. In Olivier Danvy, editor, *International Conference on Functional Programming*, pages 2–14. ACM Press, 2011.
- [Hut16] Graham Hutton. *Programming in Haskell, 2nd Edition*. Cambridge University Press, 2016.
- [KI15] Oleg Kiselyov and Hiromi Ishii. Freer monads, more extensible effects. In John H Reppy, editor, *Symposium on Haskell*, pages 94–105. ACM Press, 2015.
- [LHJ95] Sheng Liang, Paul Hudak, and Mark Jones. Monad transformers and modular interpreters. In Ron K. Cytron and Peter Lee, editors, *Symposium on Principles of Programming Languages*, pages 333–343. ACM Press, 1995.
- [Lip11] Miran Lipovača. *Learn You a Haskell for Great Good!* No Starch Press, 2011. Available online at <http://learnyouahaskell.com/>.
- [Oka99] Chris Okasaki. Red-black trees in a functional setting. *Journal of Functional Programming*, 9(4):471–477, 1999.
- [OSG98] Bryan O’Sullivan, Don Stewart, and John Goerzen. *Real World Haskell*. O’Reilly, 1998. Available online at <http://book.realworldhaskell.org/>.

A GHCi Commands

<code><statement></code>	evaluate/run <code><statement></code>
<code>:</code>	repeat last command
<code>:\{\n ..lines.. \n:\}\n}</code>	multiline command
<code>:add [*]<module> ...</code>	add module(s) to the current target set
<code>:browse[!] [[*]<mod>]</code>	display the names defined by module <code><mod></code> (!: more details; *: all top-level names)
<code>:cd <dir></code>	change directory to <code><dir></code>
<code>:cmd <expr></code>	run the commands returned by <code><expr>::IO String</code>
<code>:ctags[!] [<file>]</code>	create tags file for Vi (default: "tags") (!: use regex instead of line number)
<code>:def <cmd> <expr></code>	define command <code>:<cmd></code> (later defined command has precedence, <code>:<cmd></code> is always a builtin command)
<code>:edit <file></code>	edit file
<code>:edit</code>	edit last module
<code>:etags [<file>]</code>	create tags file for Emacs (default: "TAGS")
<code>:help, :?</code>	display this list of commands
<code>:info [<name> ...]</code>	display information about the given names
<code>:issafe [<mod>]</code>	display safe haskell information of module <code><mod></code>
<code>:kind <type></code>	show the kind of <code><type></code>
<code>:load [*]<module> ...</code>	load module(s) and their dependents
<code>:main [<arguments> ...]</code>	run the main function with the given arguments
<code>:module [+/-] [*]<mod> ...</code>	set the context for expression evaluation
<code>:quit</code>	exit GHCi
<code>:reload</code>	reload the current module set
<code>:run function [<arguments> ...]</code>	run the function with the given arguments
<code>:script <filename></code>	run the script <code><filename></code>
<code>:type <expr></code>	show the type of <code><expr></code>
<code>:undef <cmd></code>	undefine user-defined command <code>:<cmd></code>
<code>!<command></code>	run the shell command <code><command></code>

Commands for debugging

<code>:abandon</code>	at a breakpoint, abandon current computation
<code>:back</code>	go back in the history (after <code>:trace</code>)
<code>:break [<mod>] <l> [<col>]</code>	set a breakpoint at the specified location
<code>:break <name></code>	set a breakpoint on the specified function
<code>:continue</code>	resume after a breakpoint
<code>:delete <number></code>	delete the specified breakpoint
<code>:delete *</code>	delete all breakpoints
<code>:force <expr></code>	print <code><expr></code> , forcing unevaluated parts
<code>:forward</code>	go forward in the history (after <code>:back</code>)
<code>:history [<n>]</code>	after <code>:trace</code> , show the execution history
<code>:list</code>	show the source code around current breakpoint
<code>:list identifier</code>	show the source code for <code><identifier></code>
<code>:list [<module>] <line></code>	show the source code around line number <code><line></code>
<code>:print [<name> ...]</code>	prints a value without forcing its computation
<code>:sprint [<name> ...]</code>	simplified version of <code>:print</code>
<code>:step</code>	single-step after stopping at a breakpoint
<code>:step <expr></code>	single-step into <code><expr></code>
<code>:steplocal</code>	single-step within the current top-level binding
<code>:stepmodule</code>	single-step restricted to the current module
<code>:trace</code>	trace after stopping at a breakpoint

`:trace <expr>` evaluate `<expr>` with tracing on (see `:history`)

Commands for changing settings

<code>:set <option> ...</code>	set options
<code>:seti <option> ...</code>	set options for interactive evaluation only
<code>:set args <arg> ...</code>	set the arguments returned by <code>System.getArgs</code>
<code>:set prog <programe></code>	set the value returned by <code>System.getProgName</code>
<code>:set prompt <prompt></code>	set the prompt used in GHCi
<code>:set editor <cmd></code>	set the command used for <code>:edit</code>
<code>:set stop [<n>] <cmd></code>	set the command to run when a breakpoint is hit
<code>:unset <option> ...</code>	unset options

Options for `:set` and `:unset`

<code>+m</code>	allow multiline commands
<code>+r</code>	revert top-level expressions after each evaluation
<code>+s</code>	print timing/memory stats after each evaluation
<code>+t</code>	print type after evaluation
<code>-<flags></code>	most GHC command line flags can also be set here (eg. <code>-v2</code> , <code>-fglasgow-exts</code> , etc). For GHCi-specific flags, see User's Guide, Flag reference, Interactive-mode options.

Commands for displaying information

<code>:show bindings</code>	show the current bindings made at the prompt
<code>:show breaks</code>	show the active breakpoints
<code>:show context</code>	show the breakpoint context
<code>:show imports</code>	show the current imports
<code>:show modules</code>	show the currently loaded modules
<code>:show packages</code>	show the currently active package flags
<code>:show language</code>	show the currently active language flags
<code>:show <setting></code>	show value of <code><setting></code> , which is one of <code>[args, prog, prompt, editor, stop]</code>
<code>:showi language</code>	show language flags for interactive evaluation